Software Packages in Physics 0332481

Session 2: Model Solutions

Quantum Harmonic Oscillator

Part 1: The wavefunction ψ

1.1 Hints to Algorithm

1. You will first need to solve (S.2.2) for $\psi(x)$, this would involve using *Mathematica*'s **DSolve[]**. We recommend that you do not impose your boundary conditions in your application of **DSolve[]**, but later on. Also, when writing your differential equation write it *exactly* as you would when you want to solve it by hand, i.e. your equation must contain $\hbar$, m ω, and E. but when writing the symbol for energy, do not forget that **E** is a reserved by *Mathematica* for the exponential constant; instead use ESC**E**ESC, which looks the same.

▼ **Solution**

In[1]:=

```
sol = DSolve[-ℏ^2/(2 m) ψ''[x] + 1/2 m ω^2 x^2 ψ[x] == E ψ[x], ψ[x], x][[1]]
```

Out[1]=

$$\left\{\psi[x] \to e^{-\frac{m x^2 \omega}{2\hbar}} C[1]\, \mathrm{HermiteH}\left[\frac{2 E - \omega\hbar}{2\omega\hbar}, \frac{\sqrt{m}\, x \sqrt{\omega}}{\sqrt{\hbar}}\right] + e^{-\frac{m x^2 \omega}{2\hbar}} C[2]\, \mathrm{Hypergeometric1F1}\left[-\frac{2 E - \omega\hbar}{4\omega\hbar}, \frac{1}{2}, \frac{m x^2 \omega}{\hbar}\right]\right\}$$

▲ **End of Solution**

2. Exclude the unphysical solution by setting its undetermined constant to zero (*Hint: since* **DSolve[]** *outputs the solution in the form of a rule, you can use the* **Replace[]** *symbol* **/.** *to discard the unwanted part of the solution*)

▼ **Solution**

In[2]:=
```
sol1 = (ψ[x] /. sol) /. C[2] → 0
```

Out[2]=

$$e^{-\frac{m x^2 \omega}{2 \hbar}} \, C[1] \, \text{HermiteH}\left[\frac{2 E - \omega \hbar}{2 \omega \hbar}, \frac{\sqrt{m} \, x \sqrt{\omega}}{\sqrt{\hbar}}\right]$$

Notice that the boundary conditions come from two physical considerations: the fact that ψ vanishes at infinity, and that the wavefunction should be normalized to allow for a probabilistic interpretation.

From the series representation of the confluent hypergeometric function given below, one can show that the other solution is not zero at infinity, so that it does not satisfy the physical boundary conditions defining our problem (see for example Griffiths, §2.3.2).

▲ **End of Solution**

1.2 Exercises and Questions

1. Write down the differential equation and series representation for the Kummer confluent hypergeometric function ${}_1F_1(a; b; z)$.

▼ **Solution**

Solution to this exercise can be readily obtained from the *Mathematica* book! Here's is some of what is said about this function there:

The ${}_1F_1$ confluent hypergeometric function is a solution to Kummer's differential equation $z\,y'' + (b - z)\,y' - a\,y = 0$, with the boundary conditions ${}_1F_1(a; b; 0) = 1$ and $\partial[{}_1F_1(a; b; z)]/\partial z\,|_{z=0} = a/b$.

The confluent hypergeometric function can be obtained from the series expansion ${}_1F_1(a; b; z) = 1 + a\,z/b + a\,(a+1)/b\,(b+1)\,z^2/2! + \cdots = \sum_{k=0}^{\infty} (a)_k/(b)_k\, z^k/k!$

where $(symbol)_k$ is the Pochhammer symbol:

$(a)_n \equiv a(a+1)\ldots(a+n-1) = \Gamma(a+n)/\Gamma(a)$

Many special functions can be written as special cases of a hypergeometric function. To understand this, you need to look at the series solution above, and to modify it in such a manner so that its terms resemble the special function we want (for more details, and some specific examples on this, see Laham and Abdallah, *Special functions for Scientists and Engineers*, ch. 7).

▲ **End of Solution**

2. Recall that quantization is the result of imposing certain boundary conditions on the solution, and in our case this implies that the series defining the Hermite $H_n(x)$ function must be terminated, turning it into a polynomial of integral order n, so as to make the wavefunction square-integrable. By imposing this condition on your solution, show from the result of **`DSolve[]`** that energy gets quantized (use **`Solve[]`** to show that).

▼ **Solution**

```
Solve[sol1[[3, 1]] == n, E]
```

$$\left\{\left\{E \to \frac{1}{2}\ (1 + 2\ n)\ \omega\ \hbar\right\}\right\}$$

▲ **End of Solution**

3. For the solution obtained from **`DSolve[]`**, normalize the wavefunction $\psi_n(x)$ for $n = 0, 1, 2, 3, 4, 5$, and then list your result in a table of two columns the first containing the order of the wavefunction, and the second containing the normalization constant. (*Hint: Use Mathematica*'s **`Table[]`**).

Hints:

a. Assume that $\xi \equiv \sqrt{\frac{\hbar}{m\,\omega}}$, so that the functions to be normalized should have the form:

$$\psi_n \sim e^{-x^2/2\,\xi^2}\, H_n(x/\xi) \tag{S.2.1}$$

b. When using **`Integrate[]`** *to find the normalization constant you will need to use the option* **`Assumptions`**; *i.e. you need to integrate while imposing a restriction on your integrand, and this will be:* **`Re[ξ^2]>0`**. What happens if this condition is not imposed?

c. Integrate each function individually... or you will run into trouble!

▼ **Solution**

$$\psi_{n\_}[xx\_] := sol1\ //.\ \left\{\frac{2\,E - \omega\,\hbar}{2\,\omega\,\hbar} \to n,\ \frac{\sqrt{m}\,\sqrt{\omega}}{\sqrt{\hbar}} \to \frac{1}{\xi},\ \frac{m\,\omega}{\hbar} \to \xi^{-2},\ x \to xx\right\}$$

```
NormalizationConstants =
  Table[{n, C[1] /. Solve[Integrate[ψn[x]^2, {x, -∞, ∞}, Assumptions → Re[ξ^2] > 0] == 1,
      C[1]][[2, 1]]}, {n, 0, 5}];
TableForm[NormalizationConstants]
```

0	$\frac{1}{\pi^{1/4}(\xi^2)^{1/4}}$
1	$\frac{1}{\sqrt{2}\,\pi^{1/4}(\xi^2)^{1/4}}$
2	$\frac{1}{2\sqrt{2}\,\pi^{1/4}(\xi^2)^{1/4}}$
3	$\frac{1}{4\sqrt{3}\,\pi^{1/4}(\xi^2)^{1/4}}$
4	$\frac{1}{8\sqrt{6}\,\pi^{1/4}(\xi^2)^{1/4}}$
5	$\frac{1}{16\sqrt{15}\,\pi^{1/4}(\xi^2)^{1/4}}$

▲ **End of Solution**

4. Define *Mathematica* functions for each of your normalized wavefunctions, and plot them using an appropriate range. (*Hint: What provision should you make about ξ when you plot?*). Make use of **`Table[]`** and **`GraphicsArray[]`** to plot every two of the six graphs in the same cell (*but not on the same graph*).

▼ **Solution**

```
Normalizedψn_[x_] := (ψn[x] /. C[1] → NormalizationConstants[[n + 1, 2]])
```

In order to be able to plot the normalized functions, we need to give a value to ξ (in which all of the remaining undetermined constants have been compounded). We can choose this arbitrarily for the purpose of plotting, so let $\xi \equiv 1$. If we don't make a choice we get an error message telling us that the function we are plotting is not a machine-sized real number (i.e. does not evaluate to a numerical value).

```
Do[
 p[n] = Plot[Normalizedψn[x] /. ξ → 1, {x, -5, 5},
   DisplayFunction → Identity, PlotLabel → ψn], {n, 0, 5}
]


Do[
 Show[GraphicsArray[{p[n], p[n + 1]}], DisplayFunction → $DisplayFunction], {n, 0, 4, 2}
]
```



or using one output cell (this allows us to treat all plots as one unit, e.g. copy or magnify them all together):

```
plots1 = Table[{p[n], p[n + 1]}, {n, 0, 4, 2}];
Show[GraphicsArray[plots], DisplayFunction → $DisplayFunction];
```



▲ **End of Solution**

5. Is it possible to know the order of ψ directly from these plots? What observations can one make regarding the differences between wavefunctions of odd and even orders?

▼ **Solution**

From the plots above, one can observe that the wavefunction crosses the x-axis a number of times equal to the order of that wavefunction. This is an instance of the so-called ‘oscillation theorem’, which states that if the discrete eigenvalues of a 1D Schrödinger equation are placed in order of increasing magnitude

$E_1 < E_2 < ... < E$, then the corresponding eigenfunctions will occur in increasing order of their zeros, the *n*th eigenfunction having $n-1$ zeros.

By comparing the plots of wavefunctions of even and odd orders above, one can see that the wavefunctions of even order are symmetric about the origin, and that the probability amplitude of finding a particle of such a wavefunction at the origin is non-zero, while for an odd-ordered state, the wavefunction is anti-symmetric about the origin, so that the amplitude of finding the particle at the origin is zero. These things mean that even and odd orders actually describe the parity of the corresponding wavefunctions:

$$\psi_{2n}(-x) = \psi_{2n}(x)$$
$$\psi_{2n+1}(-x) = -\psi_{2n+1}(x)$$

▲ **End of Solution**

6. Using the normalized general expression for $\psi_n(x)$ (you can get this from a quantum mechanics textbook), plot $\psi_{-1}(x)$, $\psi_{-2}(x)$, $\psi_{-3}(x)$, then plot $H_{-1}(x)$, $H_{-2}(x)$, $H_{-3}(x)$. What conclusions can you make?

▼ **Solution**

```
TextBookψn_[x_] := 1/Sqrt[2^n π^(1/2) ξ n!] e^(-x^2/(2 ξ^2)) HermiteH[n, x / ξ]
```

```
pp = Table[Plot[(TextBookψn[x] /. ξ → 1), {x, -1, 1},
    PlotRange → All, DisplayFunction → Identity, PlotLabel → ψn], {n, -3, -1}];
Show[GraphicsArray[pp]];
```



Wavefunctions with negative order are all zeros due to the presence of $n!$ in the denominator. This is to be expected if we are to have a spectrum that starts with a ground state. On the other hand, the Hermite polynomials of negative order are well-defined, but are clearly not suitable for our purpose, as can be seen below:

```
Plot[Evaluate[Table[HermiteH[n, x], {n, -3, -1}]], {x, -1, 1}];
```




```
Limit[HermiteH[-1, x], x → -∞]
```

$$\text{DirectedInfinity}\left[\frac{1-i}{\sqrt{2}}\right]$$

▲ **End of Solution**

Part 2: Expectation values

2.2 Exercises and Questions

1. For $\psi_n(x)$ $(n = 0, 1, 2, 3, 4, 5)$, plot $|\psi_n(x)|^2$ on the same graph. (On your plot, you should appropriately distinguish between curves belonging to different ψ's).

▼ **Solution**

```
Plot[Evaluate[Table[Normalizedψn[x]^2 /. ξ → 1, {n, 0, 5}]],
  {x, -4, 4}, Epilog → {Text["|ψ0|^2", {.01, .6}], Text["|ψ1|^2", {1, .44}],
   Text["|ψ2|^2", {1.6, .4}], Text["|ψ3|^2", {2.2, .37}],
   Text["|ψ4|^2", {2.85, .34}], Text["|ψ5|^2", {3.4, .3}]},
  PlotStyle → Table[Dashing[{0.005 i, 0.005 i}], {i, 0, 5}], PlotRange → All];
```



▲ **End of Solution**

2. For $\psi_n(x)$ ($n = 0, 1, 2, 3, 4, 5$), make a table with the following columns: n, $\langle x\rangle$, $\langle p\rangle$, $\langle x^2\rangle$, $\langle p^2\rangle$, and from your result state your conclusions. ($\hbar$, m, and ω need to be used explicitly in your wavefunctions).

▼ **Solution**

```
TableForm[ExpectationValues =
  Simplify[Table[{n, Integrate[Simplify[Normalizedψn[x] x Normalizedψn[x]],
      {x, -∞, ∞}, Assumptions → Re[ξ^2] > 0],
     Integrate[Simplify[-i ħ Normalizedψn[x] D[Normalizedψn[x], x]],
      {x, -∞, ∞}, Assumptions → Re[ξ^2] > 0], Integrate[Simplify[
       Normalizedψn[x] x^2 Normalizedψn[x]], {x, -∞, ∞}, Assumptions → Re[ξ^2] > 0],
     Integrate[Simplify[- ħ^2 Normalizedψn[x] D[Normalizedψn[x], {x, 2}]],
      {x, -∞, ∞}, Assumptions → Re[ξ^2] > 0]}, {n, 0, 5}] /. ξ → Sqrt[ħ/(m ω)]]]
```

0	0	0	$\frac{\hbar}{2 m \omega}$	$\frac{m \omega \hbar}{2}$
1	0	0	$\frac{3 \hbar}{2 m \omega}$	$\frac{3 m \omega \hbar}{2}$
2	0	0	$\frac{5 \hbar}{2 m \omega}$	$\frac{5 m \omega \hbar}{2}$
3	0	0	$\frac{7 \hbar}{2 m \omega}$	$\frac{7 m \omega \hbar}{2}$
4	0	0	$\frac{9 \hbar}{2 m \omega}$	$\frac{9 m \omega \hbar}{2}$
5	0	0	$\frac{11 \hbar}{2 m \omega}$	$\frac{11 m \omega \hbar}{2}$

▲ **End of Solution**

3. Construct a new table from the one you obtained in the previous exercise so that it has the following columns: n, σ_p, σ_x. (*Hint: Use Mathematica's* **`Transpose[]`** *and* **`Part[]`** *to construct the new table from the old one*). What conclusions can you draw from this new table?

▼ **Solution**

```
⟨x⟩ = Transpose[ExpectationValues][[2]];
⟨x^2⟩ = Transpose[ExpectationValues][[4]];
⟨p⟩ = Transpose[ExpectationValues][[3]];
⟨p^2⟩ = Transpose[ExpectationValues][[5]];
```

```
σx = ⟨x^2⟩ - ⟨x⟩^2
```

$$\left\{\frac{\hbar}{2\,m\,\omega}, \frac{3\,\hbar}{2\,m\,\omega}, \frac{5\,\hbar}{2\,m\,\omega}, \frac{7\,\hbar}{2\,m\,\omega}, \frac{9\,\hbar}{2\,m\,\omega}, \frac{11\,\hbar}{2\,m\,\omega}\right\}$$

```
σp = ⟨p^2⟩ - ⟨p⟩^2
```

$$\left\{\frac{m\,\omega\,\hbar}{2}, \frac{3\,m\,\omega\,\hbar}{2}, \frac{5\,m\,\omega\,\hbar}{2}, \frac{7\,m\,\omega\,\hbar}{2}, \frac{9\,m\,\omega\,\hbar}{2}, \frac{11\,m\,\omega\,\hbar}{2}\right\}$$

```
σx σp
```

$$\left\{\frac{\hbar^2}{4}, \frac{9\,\hbar^2}{4}, \frac{25\,\hbar^2}{4}, \frac{49\,\hbar^2}{4}, \frac{81\,\hbar^2}{4}, \frac{121\,\hbar^2}{4}\right\}$$

```
TableForm[Transpose[{Transpose[ExpectationValues][[1]], σx, σp, Sqrt[σx σp]}]]
```

0	$\frac{\hbar}{2\,m\,\omega}$	$\frac{m\,\omega\,\hbar}{2}$	$\frac{\sqrt{\hbar^2}}{2}$
1	$\frac{3\,\hbar}{2\,m\,\omega}$	$\frac{3\,m\,\omega\,\hbar}{2}$	$\frac{3\sqrt{\hbar^2}}{2}$
2	$\frac{5\,\hbar}{2\,m\,\omega}$	$\frac{5\,m\,\omega\,\hbar}{2}$	$\frac{5\sqrt{\hbar^2}}{2}$
3	$\frac{7\,\hbar}{2\,m\,\omega}$	$\frac{7\,m\,\omega\,\hbar}{2}$	$\frac{7\sqrt{\hbar^2}}{2}$
4	$\frac{9\,\hbar}{2\,m\,\omega}$	$\frac{9\,m\,\omega\,\hbar}{2}$	$\frac{9\sqrt{\hbar^2}}{2}$
5	$\frac{11\,\hbar}{2\,m\,\omega}$	$\frac{11\,m\,\omega\,\hbar}{2}$	$\frac{11\sqrt{\hbar^2}}{2}$

In general, we can verify analytically that for the harmonic oscillator we have $\sigma_x\,\sigma_p = \left(n + \frac{1}{2}\right)\hbar$, i.e. the product $\sigma_x\,\sigma_p$ increases with the quantum number n, which means that a simultaneous specification of coordinate and momentum can be made with greater accuracy for states with lower quantum number n than for those with larger n, with the minimum uncertainty occurring at the ground state defining a Gaussian wavepacket.

▲ **End of Solution**

Part 3: Classical vs. Quantum Mechanical Harmonic Oscillator

3.1 Exercises and Questions

1. Solve problem 1.6 from Pain's book.

2. Using the result for the eigenvalues of energy obtained in part 1 above, and the result from problem 1.6 from Pain, and using the fact that for the classical oscillator $E = \frac{1}{2} m x_0^2 \omega^2$, show that the classical analogue of $|\psi_n(x)|^2 \equiv q_n(x)$ is given by:

$$p_n(x) = \frac{1}{\pi \sqrt{(2n+1)\xi^2 - x^2}} \tag{S.2.2}$$

where ξ is as defined in part 1, above.

▼ **Solution**

The classical motion takes place between the turning points at which $E(x) = V(x)$, i.e. at $x = x_0 = \pm\sqrt{2E/m\omega^2}$. The probability $p(x)\,dx$ of finding the particle in the interval dx around x equals the fraction of time the particle spends at that location:

$$p(x)\,dx = \frac{1}{T}\frac{2\,dx}{v} = \frac{dx}{\pi\sqrt{x_0^2 - x^2}}$$

where $T = 2\pi/\omega$ is the period of oscillation, and v is the speed of the oscillator. Now, since quantum mechanically E is quantized according to $E_n = \left(n + \frac{1}{2}\right)\hbar\omega$, we may rewrite the probability as:

$$p_n(x)\,dx = \frac{dx}{\pi\sqrt{(2n+1)\xi^2 - x^2}}$$

QED

▲ **End of Solution**

3. On the same graph, and for $n = 0$ plot the quantum probability $q_0(x)$ and the classical probability function $p_0(x)$. On a second graph do the same, but this time for $n = 20$. What do you conclude?

▼ Solution

```
p[n_, x_] := 1/(π Sqrt[2 n + 1 - x^2])
q[n_, x_] := TextBookψn[x]^2 /. ξ → 1
```

$$p[n\_, x\_] := \frac{1}{\pi\sqrt{2n+1-x^2}}$$

```
p0 =
  Plot[{p[0, x], q[0, x]}, {x, -1, 1}, AxesLabel → {x, "Probability density for n=0"},
   Epilog → {Text[classical, {.8, 1.6}], Text[quantal, {.18, .7}]}];
```




```
Solve[2 * 20 + 1. - x^2 == 0, x]
```

```
{{x → -6.40312}, {x → 6.40312}}
```

```
p20a = Plot[q[20, x], {x, -10, 10}, DisplayFunction → Identity];
p20b = Plot[p[20, x], {x, -6.4, 6.4}, DisplayFunction → Identity];
p20 = Show[{p20a, p20b}, AxesLabel → {x, "Probability density for n=20"},
   Epilog → {Text[classical, {7, .255}], Text[quantal, {8.5, .05}]},
   DisplayFunction → $DisplayFunction];
```



We notice that for $n = 0$ the behaviours of the classical and quantal oscillators are quite different, however, for $n = 20$, apart from the rapid oscillations in the quantal case, the bahaviour in both cases is quite similar. This observation is a verification for the correspondence principle in the case of the harmonic oscillator, so that in the limit of large values of the quantum number n we recover the classical behaviour as expected.

▲ **End of Solution**